
Summary 

This manual describes the PEARLNEQ v4 sofware package . This package can estimate long-
term sorption parameters using results of aged-sorption studies with soil. using a submodel 
for sorption and transformation that is identical to the submodel used for that purpose in the 
FOCUS_PEARL v3.3.3. 
 
The submodel assumes two types of sorption sites: equilibrium sites and non-equilibrium sites. 
The sorption isotherms for bot sites are described with Freundlich equations. The content 
sorbed at the equilibrium site is assumed to be continuously at equilibrium and the content 
sorbed at the non-equilibrium site is described with a pseudo first-order sorption rate equation. 
The transformation rate in soil is assumed to be proportional to the amount in the liquid phase 
plus the amount sorbed at the equilibrium site. So the content sorbed at the non-equilibrium 
site is not subject to transformation. 
  
The mathematical equations describing the submodel are solved via a FORTRAN programme. 
An additional FORTRAN programme generates the necessary input files for the PEST 
optimisation package. Instructions are given how to obtain optimized parameters using an 
example dataset and, subsequently instructions are given how to obtain parameters using 
your own data. 

1 Introduction 

This document describes a PEARLNEQ-PEST combination, which can be used to estimate the 
parameters for long-term sorption kinetics in the PEARL model on the basis of an incubation 
experiment for a certain soil and a certain pesticide. The combination provides also the 
transformation half-life at reference temperature (when long-term sorption kinetics are included 
in PEARL, the definition of this half-life changes so it has to be recalculated; see Boesten and 
van der Linden, 2001). If the incubation experiment has been carried out at multiple 
temperatures, the Arrhenius activation energy for the transformation rate in soil can be 
optimised simultaneously. 

2 Precautionary remark 

This PEARLNEQ-PEST software tool should be seen as an introduction to fitting results of 
experiments on long-term sorption kinetics to the sorption submodel used in the PEARL 
model. The tool shows you how PEST can be coupled to a fortran programme that contains 
this PEARL sorption submodel (i.e. PEARLNEQ.EXE) but it should not be seen as a ready-to-use 
tool. The tool provides you with example input files for the PEST optimisation package and 
shows you how to organise this optimisation. We had to make a number of assumptions for 
generating these PEST input files (e.g. upper and lower bounds of parameters, weighing 
factors for each measurement, etc. etc.). We do not claim that these assumptions are 
defensible for your problem; they are our best guesses but they may not be appropriate for 
your problem. It is your responsibility to check the appropriateness of the result obtained. We 
do not accept any responsibility for use of PEARLNEQ. 

3 Description of the incubation experiment 

The PEARLNEQ-PEST tool can be used to fit the results of the following experiment. A number 
of jars is filled with soil. Each jar contains the same mass of moist soil. At the start of the 
experiment the same initial mass of pesticide is added to the moist soil in all jars. The jars are 
incubated at a fixed temperature (or at a few temperatures). At certain time points the 
remaining total amount of pesticide is measured via an extraction with organic solvent. At the 
same time the concentration in the liquid phase of the moist soil is measured. The liquid phase 
can be collected by centrifuging the moist soil over a filter. As an alternative for centrifuging, a 
desorption experiment can be carried out by adding a certain volume of water and subsequent 
shaking for about 24 h. 
 
It is assumed that additionally an adsorption isotherm with an equilibration time of about 24 h 
has been measured for the same soil and pesticide. 

4 Theoretical background 

PEARLNEQ assumes a Freundlich two-site sorption submodel: one site for equilibrium sorption 
and the second site for long-term sorption kinetics. The operational definition for the 
equilibrium sorption sites is that they have reached equilibrium after about 24 h shaking of a 
well-stirred suspension of the soil in water. The long-term sorption sites do not reach 
equilibrium within 24 h. PEARLNEQ assumes first order degradation kinetics for the molecules 
present in liquid phase and sorbed to the equilibrium site; however, molecules sorbed on the 
kinetic site are assumed not to degrade. This conceptual model is presented in Figure 1. The 
submodel for sorption and degradation kinetics used in PEARLNEQ can be described as 
follows (Leistra et al., 2001): 

KF,EQ = mOM KOM,EQ      (6) 
 
 
where: 
Mp = initial total mass of pesticide in each jar (μg), acronym MasIni  
V = the volume of water in the soil incubated in each jar (mL), acronym VolLiq 
Ms = the mass of dry soil incubated in each jar (g), acronym MasSol 
cL = concentration in the liquid phase (μg/L), acronym ConLiq 
cL,R = reference concentration in the liquid phase (μg/L), acronym ConLiqRef 
XEQ = content sorbed at equilibrium sites (μg/g) 
XNE = content sorbed at non-equilibrium sites (μg/g) 
KF,EQ = equilibrium Freundlich sorption coefficient (mL/g), acronym CofFreEql 
KF,NE = non-equilibrium Freundlich sorption coefficient (mL/g), acronym CofFreNeq 
N = Freundlich exponent (-),acronym ExpFre 
kd = desorption rate coefficient (d-1), acronym CofRatDes  
fNE = a factor for describing the ratio between the equilibrium and non-equilibrium Freundlich 
coefficients (-), acronym FacSorNeqEql  
kt = degradation rate coefficient (d-1) 
mOM = mass fraction of organic matter in the soil (kg/kg), acronym CntOm 
KOM,EQ = coefficient of equilibrium sorption on organic matter (mL/g), acronym KomEql 

PEARLNEQ does not use the transformation rate coefficient (kt) as input parameter, but the 
half-life at reference temperature (acronym DT50Ref, dt50). They are related as follows 
(assuming first order kinetics): 
 
dt50 = ln (2) / kt       (7) 
 
The effect of soil temperature on the transformation rate coefficient in soil is described by the 
Arrhenius equation: 

where 
fT = the multiplication factor for the rate coefficient (-) 
E = Arrhenius activation energy (kJ/mol) 
T = temperature of the soil (K) 
TREF = the reference temperature for the specified DT50 (K) 
R = the gas constant (kJ mol-1 K-1 ). 
 

 
Figure 1. Conceptual representation of the PEARLNEQ model. 
 
Often no concentration measurements in the soil pore water are available but instead at each 
sampling point in time a certain volume of water (usually a CaCl2 solution) is added to soil and 
the suspension is shaken for about 24 h after which the concentration in the supernatant is 
measured. In such a case the fit has to be based on these concentration measurements in the 
supernatant of the soil-water suspension. 
 
This is simulated in PEARLNEQ as follows: 
(A) it is assumed that full equilibrium is reached for the equilibrium sorption site during the 
desorption experiment (i.e., shaking for 24 h) 
(B) it is assumed that desorption from the non-equilibrium sorption site can be ignored during 
the desorption experiment. 

Assumption A is justifiable because this is exactly the operational definition of the equilibrium 
sorption site. Assumption B is justifiable because desorption coefficients for long-term kinetics 
are usually in the order of 0.01 d-1, which implies that amounts desorbed within 1 day are 
negligibly small. 
 
Using these assumptions, the concentration in the liquid phase of the supernatant after 
desorption can be estimated by stating that (i) the total content of substance in the moist soil 
and the soil-water suspension have to be equal, and (ii) the content sorbed at the non-
equilibrium sites in the moist soil and in the soil-water suspension are equal. Using Equation 1 
then results in the following equation 
 

( ) ( ) ( )NESUSEQsSUSLADDNEMSEQsMSL XXMcVVXXMcV +++=++ ,,,,   (9)  

 
where:  
the subscript MS indicates the moist-soil system 
the subscript SUS indicates the soil-water suspension system and 
VADD = volume of liquid (usually CaCl2 solution) added to the soil at each sampling point just 
before starting the 24 h desorption experiment (mL), acronym VolLiqAdd. 
 
At each sampling point in time, Equation 9  can be rewritten (using Eqn 2) into an equation that 
contains only one unknown variable, i.e. the concentration in the liquid phase of the soil 
suspension (cL,SUS). PEARLNEQ provides as output always the concentration in the soil-water 
suspension as a function of time. If VADD = 0, then this implies that the concentration in the 
moist soil is given. 
 
PEARLNEQ solves the set of Eqn 1 to Eqn 9 numerically using Euler’s method for integration 
of the state variables Mp and XNE. The time step for integration can be set as an input 
parameter (we recommend 0.01 d). The concentration in the liquid phase is calculated via an 
iteration procedure as described in Appendix 4 of FOCUS (2006). Appendix 3 shows a test of 
the PEARLNEQ results against an analytical solution for the case of a linear isotherm (N=1), 
indicating good correspondence between numerical and analytical results. 

5 Fitting procedure for parameters with PEST 

The provided package assumes that the following variables need to be optimized: 
• the initial mass of the pesticide (MasIni) 
• the ratio between the equilibrium and non-equilibrium Freundlich coefficients 

(FacSorNeqEql) 
• the desorption rate coefficient (CofRatDes) 
• the half-life at reference temperature (DT50Ref) 
• the molar activation energy (MolEntTra); this can only be optimized if the experiment has 

been carried out at multiple temperatures.  
It is assumed that all other variables are known. 
 
The provided package assumes that the measurements that are fitted, consist for each point 
in time of 
• a mass of pesticide in μg 
• a concentration in liquid phase in μg/mL. 
 
PEST needs a number of input parameters for the fitting procedure (e.g. upper and lower 
bounds of parameters, weighing factor for each measurement, etc. etc.). Our experience is 
that the weighing factor for each measurement is the most important input parameter.  
 
Therefore we offer two options for weighing: 
‘equal’ which gives a weight of 1.0 to all observations (so equal weights)  
‘inverse’ which gives a weight that is proportional to the inverse of the observed value. 
If the observed value is zero, the weight is set equal to 1.0 in any case. 
 
The option ‘equal’ implies that high observed values get more weight than low observed 
values. As described above, the fitting procedure considers two quantities: mass of pesticide 
and the concentration in the liquid phase. This may lead to completely different weights for 
these two types of quantities.  
E.g. if the mass is initially 50 μg and the concentration in the liquid phase is in the order of 1 
μg/mL, then the fitting procedure will be completely dominated by the decline of the mass of 
pesticide. So if the option ‘equal’ is used, the user should choose a mass of solid phase such 
that the values of the mass of pesticide in μg should be in the same order of magnitude as the 
concentration in liquid phase in μg/mL. 
 
The option ‘inverse’ implies that each measurement gets more or less equal weight for the 
parameter estimation. This ‘inverse’ option gave the best results in a few tests. However, we 
do not claim that this is the best choice for your dataset nor do we claim that the other PEST 
input parameters are the best choice for your dataset.  
 
The provided package cannot handle duplicate or triplicate observations for each point in 
time. Within PEST there are procedures available to do so but these have not yet been 
implemented in PEARLNEQ. So the user has to average first all measurements for each point 
in time.  

6 Installation of PEARLNEQ  

PEARLNEQ is distributed in a zip file. Unzip the file and specify a path (e.g. c:\pearlneq). Be 
sure there is no space within the specified path, because this will cause of failure. The 
package contains four directories, i.e. Neq_Bin, Pest, Neq_fortran_source_files and 
Neq_Example. 
• The Neq_Bin directory contains the PEARLNEQ executables, PEARLNEQ.EXE and 

PEARLMK.EXE. 
• The PEST optimisation software is available in the Pest directory. As PEST is now 

available freeware (http://www.sspa.com/pest), we included the latest version as of 29-
10-2003 (version 7.0.1.). Separate installation of PEST is not necessary. 

• The Neq_fortran_source_files directory contains the fortran source files used to 
generated the PEST input files and the programme that calculates the sorption kinetics 

• The Neq_Example directory contains results from an example study, as described in the 
FOCUS PEARL user manual, version 1.1.1 § 3.2.10 (page 51-54). 

7 Running the example 

The following steps must be followed. 
 
1 Run the example, to check if everything works and get experience with the system. 

Go to the Neq_Example directory, and run the example, example.bat. 
• The batch file will first call PEARLMK. This pre-processing program generates the 

input files for PEST, i.e. example.pst, example.tpl and example.ins (see Figure 2, 
RunId = “example”) 

• Then, the optimisation starts. PEST calls PEARLNEQ several times (see Figure 2; 
in the example 39 times). 

• If you get an error message after the first step (PEARLMK), type control-break to 
stop the process and check the error messages available in the example.err file. 

 
2 After successful optimisation, read the results from the file example.rec. Choose 

“select the program from the list” and open with Notepad. The relevant results, 
including parameter values, 95% confidence intervals and correlation matrices can be 
found at the end of this file (Section OPTIMISATION RESULTS, see Appendix 2). The 
meaning of the short acronyms in this rec-file is as follows: 
fsne =  FacSorNeqEql 
crd = CofRatDes 
dt50 = DT50Ref 
masini =  MasIni 
met = MolEntTra.  
PEST also generates parameter sensitivity files etc. Details can be found in the PEST 
manual, which is available in the PEST subdirectory of the package. 

 
3 If you encounter errors during the second step, you can try running PEARLNEQ 

directly. PEARLMK has created a file example.neq (in …\Neq_Example) which is the 
input file for pearlneq. You can run PEARLNEQ by typing “..\Neq_Bin\pearlneq 
example” in a DOS-box.  

 
4 PEARLNEQ will create an output file (example.out) and a log file (example.log). The 

output files are self-explaining. The output file contains the result of the last run which 
is in PEST by definition the run with the optimised parameters. 

 
5 You can use the XyWin program to make a graph. (After running the example this will 

be done automatically.) XyWin is available in the neq_bin\xypearl directory of 
PEARLNEQ:  you can also get the graph via typing in a DOS box: 
“..\Neq_Bin\xypearl\xywin –j example.job –W l”.  However, XyWin is not userfriendly so 
we recommend to use the results from the output file (example.out) as the source for 
the best fit and create graphs with software of your own choice. 

RunId.mkn

PearlMk

RunId.tpl RunId.pstRunId.ins

RunId.neq

Pest

PearlNeq

RunId.out

RunId.rec

If Converged

 
 
Figure 2.  Dataflow diagram for the PEARLNEQ-PEST combination. The acronym RunId is 
“example” for the example provided.  

8 Run PEARL_Neq with your own data 

1 We assume that you have carried out an appropriate incubation experiment as described 
before. The first step of optimising your own data consists of editing the file 
example.mkn, which can be found in the example subdirectory of the PEARLNEQ di-
rectory. Open the file with Notepad. Please make a copy of this file before editing. Make 
sure there is no space in the new name. This will give an error. An example of this input 
file is listed in Appendix 1. The following parameters must be provided: 

 TimEnd (d): The duration of the incubation experiment. 
 MasSol (g): The mass of dry soil incubated in each jar. 
 VolLiqSol (mL): Volume of liquid in the moist soil during incubation. 
 VolLiqAdd (mL): Volume of liquid added to the soil after incubation (i.e. the amount of 

liquid added to perform a conventional desorption equilibrium experiment). 
 CntOm (kg.kg-1): Mass fraction of organic matter in the soil. 
 ConLiqRef (mg L-1): Reference concentration in the liquid phase. 
 ExpFre (-): Freundlich exponent; use value taken from adsorption isotherm measured 

for this pesticide-soil combination 
 KomEql (L kg-1): coefficient of equilibrium sorption on organic matter; use value taken 

from adsorption isotherm measured for this pesticide-soil combination; in case you 
have no organic matter content of the soil, set the organic matter to 1.0 and specify 
the measured Freundlich equilibrium coefficient (see Eqn 6) 

 MasIni (μg): The initial total mass of pesticide in each jar. In contrast to initial  versions 
of PEARLNEQ, this parameter will be optimised. There is no default value for this 
parameter. 

 FacSorNeqEql (-): factor describing the ratio fNE = KF,NE/KF,EQ as defined by Eqn 4. This 
parameter will be optimised, but you have to specify an initial guess here. The default 
value is 0.5. 

 CofRatDes (d-1): the desorption rate coefficient. This parameter will be optimised, but 
you have to specify an initial guess here. The default value is 0.01 d-1. 

 DT50Ref (d): the transformation half-life under reference conditions, applying to the 
equilibrium domain. This parameter will be optimised, but you have to specify an initial 
guess here. As a default value, you can use the ‘classical’ half-life, which applies to 
the total soil system (i.e. the equilibrium domain + the non-equilibrium domain). 

 TemRefTra (C): The reference temperature, for which the half-life will be provided (set 
to incubation temperature if data for only one temperature are available and set to 
20o C if you have data for multiple temperatures). 

 MolEntTra (kJ mol-1): the molar enthalpy of transformation. This parameter will be 
optimised if you have carried out the experiment at multiple temperatures; otherwise 
it is a model-input. In any case you have to specify a value (e.g. 60 kJ mol-1) which will 
be used as an initial guess in case of data for more than one temperature.  

 table Tem (C): List of temperatures at which the incubation experiment has been car-
ried out. One temperature is OK if only data for one temperature are available. 

 table Observations: List of observations. The first column contains the time (d), the 
second column the temperature, column 3 contains the total mass of pesticide in the 
system (μg), column 4 contains the concentration of pesticide (μg mL-1) measured in 
the pore water of moist soil (then VolLiqAdd = 0) or in the water phase after a 
desorption experiment (in which case VolLiqAdd is not zero) and column 5 contains 
the characters ‘OBS’. 

 option Opt_weights: options for weights. Two options for weighing are offered: ‘equal’ 
which gives a weight of 1.0 to all observations (so equal weights) and ‘inverse’ which 
gives a weight that is proportional to the inverse of the observed value; if the 
observed value is zero, the weight is set equal to 1.0 in any case; you can inspect the 
weigths in the ‘pst’ file. 

 
2. Modify the contents of the example.bat file (with right mouse button): replace “example” 

everywhere it occurs by the name of the copied input file and delete last line of 
the file (which would generate the graph). Repeat step 1-5 of chapter 7. 

 
3. If the optimization is not succesfull, you can try re-running PEARLNEQ with different initial 

guesses of MasIni, DT50Ref, FacSorNeqEql and CofRatDes. 

9 Concluding remark 

While using PEARLNEQ, we noticed that very regularly the results depend on the initial 
guesses of the parameters. Therefore we advise you to perform always a number of runs with 
different initial guesses. We advise you also to analyse the results very carefully, especially the 
95% confidence intervals of your parameters.  If the interval is wide for a certain parameter, 
this indicates that the estimated variable is very uncertain. As a consequence it is usually not 
meaningful to use it any further in the risk assessment. 

Appendix 1  Example input file 

Example input file “example.mkn” 

*----------------------------------------------------------------------------------------- 

* STANDARD FILE for pearlmk version 4 

* Program to fit the half-life, activation energy and parameters for long-term sorption  

* kinetics of pesticides in soil 

* 

* This file is intented for use with the PEST program (Doherty et al., 1991). 

* Please see manual of PEARLNEQ 

* 

* (c) RIVM/MNP/Alterra 2003, 2005, 2006, 2007 

*----------------------------------------------------------------------------------------- 

 

* Model control 

Yes              ScreenOutput 

0.0              TimStart          (d)         Start time of experiment 

500.0            TimEnd            (d)         End time of experiment 

0.01             DelTim            (d)         Time step of Euler's integration procedure  

 

* System characterization 

54.64            MasIni            (ug)             Initial guess of initial mass 

45.36            MasSol            (g)              Mass of soil in incubation jar 

6.64             VolLiqSol         (mL)             Volume of liquid in the moist soil 

0.0              VolLiqAdd         (mL)             Volume of liquid ADDED 

0.047            CntOm             (kg.kg-1)        Organic matter content 

 

* Sorption parameter 

1.0              ConLiqRef         (mg.L-1)       Reference liquid concentration 

0.87             ExpFre            (-)            Freundlich exponent 

2.1              KomEql            (L.kg-1)       Coefficient for equilibrium sorption 

0.5              FacSorNeqEql      (-)            Initial guess of ration KfNeq/KfEql 

0.01             CofRatDes         (d-1)          Initial guess of desorption rate constant 

 

* Transformation parameters 

14.00            DT50Ref           (d)       Initial guess of half-life at ref. temperature 

20.0             TemRefTra         (C)             Reference temperature 

110.0            MolEntTra         (kJ.mol-1)      Initial guess of molar activation energy 

 

* Temperature at which the incubation experiments have been carried out 

table Tem (C) 

1  5.0 

2  15.0 

end_table 

 

* Provide the results of the measurements 

* Tim  Tem  Mas       ConLiq 

* (d)  (C)  (ug)      (ug/mL) 

table Observations 

2      5    51.6300   5.7285       OBS 

10     5    50.5900   5.0560       OBS 

42     5    46.0200   3.6635       OBS 

87     5    38.6100   2.9320       OBS 

157    5    32.8150   1.9280       OBS 

244    5    25.8700   1.4650       OBS 

358    5    20.3150   0.8820       OBS 

451    5     9.4250   0.6015       OBS 

2     15    51.3300   5.8955       OBS 

6     15    47.3950   4.4425       OBS 

10    15    45.0650   3.9510       OBS 

42    15    23.1400   1.6470       OBS 

87    15    10.8950   0.6710       OBS 

157   15     3.1350   0.1525       OBS 

244   15     1.4400   0.0305       OBS 

358   15     0.4500   0.0000       OBS 

451   15     0.1500   0.0000       OBS 

end_table 

 

* Option for weights of observations: 

*'equal' gives equal weights to all measurements 

*'inverse' gives weigth equal to inverse value of each measurement (if measurement is zero 
then weight is 1.0) 

inverse    Opt_weights 

Appendix 2  Results of the default example 

Results (taken from last section of REC-file) 
 
These are the results of the defaultexample, provided with the package. 
 
                            OPTIMISATION RESULTS 
 
Parameters -----> 
 
Parameter        Estimated         95% percent confidence limits 
                 value             lower limit       upper limit 
 fsne           0.600286           0.408681          0.791892     
 crd            1.226317E-02       9.886449E-03      1.463989E-02 
 dt50            13.0062            11.5100           14.5024     
 masini          56.0664            52.1243           60.0084     
 met             108.755            102.806           114.705     
 
Note: confidence limits provide only an indication of parameter uncertainty. 
      They rely on a linearity assumption which  may not extend as far in  
      parameter space as the confidence limits themselves - see PEST manual. 
 
See file EXAMPLE.SEN for parameter sensitivities. 
 
Observations -----> 
 
Observation      Measured       Calculated     Residual       Weight     Group 
                 value          value 
 o1              51.6300        55.5316       -3.90156       1.9000E-02  no_name      
 o2              5.72850        5.39742       0.331082       0.1750      no_name      
 o3              50.5900        53.4684       -2.87844       2.0000E-02  no_name      
 o4              5.05600        5.08081      -2.480823E-02   0.1980      no_name      
 o5              46.0200        46.2635      -0.243511       2.2000E-02  no_name      
 o6              3.66350        4.06320      -0.399703       0.2730      no_name      
 o7              38.6100        38.2669       0.343149       2.6000E-02  no_name      
 o8              2.93200        3.09390      -0.161897       0.3410      no_name      
 o9              32.8150        29.0458        3.76922       3.0000E-02  no_name      
 o10             1.92800        2.16483      -0.236828       0.5190      no_name      
 o11             25.8700        20.9646        4.90536       3.9000E-02  no_name      
 o12             1.46500        1.47675      -1.175005E-02   0.6830      no_name      
 o13             20.3150        13.8380        6.47702       4.9000E-02  no_name      
 o14            0.882000       0.932184      -5.018437E-02    1.134      no_name      
 o15             9.42500        9.90684      -0.481840       0.1060      no_name      
 o16            0.601500       0.648450      -4.694962E-02    1.663      no_name      
 o17             51.3300        53.3843       -2.05426       1.9000E-02  no_name      
 o18             5.89550        5.17830       0.717199       0.1700      no_name      
 o19             47.3950        48.4369       -1.04190       2.1000E-02  no_name      
 o20             4.44250        4.62304      -0.180537       0.2250      no_name      
 o21             45.0650        43.9959        1.06909       2.2000E-02  no_name      
 o22             3.95100        4.12910      -0.178099       0.2530      no_name      
 o23             23.1400        21.3140        1.82600       4.3000E-02  no_name      
 o24             1.64700        1.70735      -6.034768E-02   0.6070      no_name      
 o25             10.8950        8.98367        1.91133       9.2000E-02  no_name      
 o26            0.671000       0.544736       0.126264        1.490      no_name      
 o27             3.13500        3.26320      -0.128203       0.3190      no_name      
 o28            0.152500       0.131795       2.070505E-02    6.557      no_name      
 o29             1.44000        1.24259       0.197406       0.6940      no_name      
 o30            3.050000E-02   3.866613E-02  -8.166130E-03    32.79      no_name      
 o31            0.450000       0.399623       5.037676E-02    2.222      no_name      
 o32             0.00000       1.077987E-02  -1.077987E-02    1.000      no_name      
 o33            0.150000       0.162574      -1.257429E-02    6.667      no_name      
 o34             0.00000       4.019330E-03  -4.019330E-03    1.000      no_name      

See file EXAMPLE.RES for more details of residuals in graph-ready format. 
 
See file EXAMPLE.SEO for composite observation sensitivities. 
 
 
Objective function -----> 
 
  Sum of squared weighted residuals (ie phi)                =  0.4297     
 
Correlation Coefficient -----> 
 
  Correlation coefficient                                   =  0.8997     
 
 
Analysis of residuals -----> 
 
  All residuals:- 
     Number of residuals with non-zero weight               =    34 
     Mean value of non-zero weighted residuals              =  1.3161E-02 
     Maximum weighted residual [observation "o13"]          =  0.3174     
     Minimum weighted residual [observation "o30"]          = -0.2677     
     Standard variance of weighted residuals                =  1.4816E-02 
     Standard error of weighted residuals                   =  0.1217     
 
     Note: the above variance was obtained by dividing the objective  
     function by the number of system degrees of freedom (ie. number of  
     observations with non-zero weight plus number of prior information  
     articles with non-zero weight minus the number of adjustable parameters.) 
     If the degrees of freedom is negative the divisor becomes  
     the number of observations with non-zero weight plus the number of  
     prior information items with non-zero weight. 
 
 
Parameter covariance matrix -----> 
 
                 fsne         crd          dt50        masini        met      
fsne           8.7787E-03   8.4444E-05  -5.6502E-02   6.4073E-02   0.1016     
crd            8.4444E-05   1.3507E-06  -4.5078E-04   6.0200E-04   6.0023E-04 
dt50          -5.6502E-02  -4.5078E-04   0.5353      -0.5756       -1.463     
masini         6.4073E-02   6.0200E-04  -0.5756        3.716      -0.8663     
met            0.1016       6.0023E-04   -1.463      -0.8663        8.463     
 
 
Parameter correlation coefficient matrix -----> 
 
                 fsne         crd          dt50        masini        met      
fsne            1.000       0.7755      -0.8242       0.3548       0.3727     
crd            0.7755        1.000      -0.5301       0.2687       0.1775     
dt50          -0.8242      -0.5301        1.000      -0.4081      -0.6872     
masini         0.3548       0.2687      -0.4081        1.000      -0.1545     
met            0.3727       0.1775      -0.6872      -0.1545        1.000     
 
 
Normalized eigenvectors of parameter covariance matrix -----> 
 
              Vector_1     Vector_2     Vector_3     Vector_4     Vector_5    
fsne           1.4335E-02  -0.9842       0.1744      -2.2879E-02   1.1198E-02 
crd           -0.9999      -1.4279E-02   1.5744E-03  -1.9999E-04   6.4642E-05 
dt50           9.4905E-04  -0.1745      -0.9453       0.2229      -0.1618     
masini         7.8350E-05  -1.4702E-02  -0.2006      -0.9685      -0.1467     
met            7.0894E-05  -1.9859E-02  -0.1889      -0.1084       0.9758     
 
Eigenvalues -----> 
 
               4.7821E-07   1.7663E-03   0.1313        3.753        8.837      

Appendix 3  Comparison between an analytical solution and 
PearlNeq 

In this appendix an analytical solution for the remaining mass of pesticide is compared with the PearlNeq 
solution (appendix 4). The system properties were: 
 
* Mass of dry soil (MasSol) (g)    1.0000 
* Volume of water in moist soil (VolLiqsol) (mL)  0.2000 
* Volume of water added (VolLiqAdd (mL)   0.0000 
* Initial mass of pesticide (MasIni) (ug)   10.0000 
* Reference concentration (ConLiqRef) (ug.mL-1)  1.0000 
* Equilibrium sorption coefficient (CofFreEql) (mL.g-1) 1.0000 
* Non-equili. sorption coefficient (CofFreNeq) (mL.g-1) 0.5000 
* Freundlich exponent (ExFre) (-)    1.0000 
* Desorption rate coefficient (CofRatDes) (d-1)  0.0100 
* Half-life transformation (DT50Ref) (d)   69.3000 
* Reference temperature (TemRefTra) (K)   293.1500 
 
The analytical solution was taken from Appendix 4 of FOCUS (2006).  
 
The figure shows that the PearlNeq solution coincides very well with the analytical solution. 

Appendix 4  Listing of Fortran program PearlNeq 

program PearlNeq 
!========================================================================================= 
! 
! PEARLNEQ program - simulates pesticide behaviour in a closed incubation system assuming 
!                    a two-site Freundlich sorption submodel and first-order transformation 
!                    kinetics 
!  
! History 
! ------- 
! 1.0.1 Initial version for Pearl 1.1 
! 1.1.1 Version compatible with FOCUS Pearl 1.1.1 
! 2.2.2 Version compatible with FOCUS Pearl 2.2.2 
! 3.3.3 Version compatible with FOCUS Pearl 3.3.3 
! Version 4 of 10 May 2007 
!========================================================================================= 
 
   use Sishell                 ! General routines 
   use CompilerSpecific        ! Compiler specific statements 
 
   implicit none 
 
   character (len=LineLength) :: Path 
 
   integer :: T,Steps,StepsToPrint 
 
   double precision, parameter :: RGas=8.31432d0   ! Molar gas constant 
   double precision, parameter :: TimeStart=0.d0   ! Start time 
   double precision, parameter :: DelTimPrint=1.d0 ! Print time step 
 
   double precision :: CntOm,CofFreEql,CofFreNeq,CofRatDes,CofRatTra,DelTim, & 
   DT50Ref,ExpFre,Mas,MasEql,MasIni,MasSol,VolLiqAdd,MolEntTra,TimeEnd,& 
   
VolLiqSol,VolLiqSus,XNeq,ConLiqRef,Tim,Tem,FacTem,TemRefTra,KomEql,FacSorNeqEql,ConPor,ConS
us 
 
   type (TableType) :: TemTab 
 
   save 
 
   IOMode = IOMode_Full 
   ShowScreen = .false. 
 
   ! Initial part of program 
   !------------------------ 
 
   ! Set the model stamp (version numbers etc) 
   Call SetModelStamp () 
 
   ! Open the input file 
   call InitCh (Path) 
   Call OpenPearlNeqFiles(Path) 
 
   ! System properties 
   !------------------ 
 
   ! Initial mass of pesticide 
   call GetInput (MasIni,'MasIni','(ug)',Valmin=0.d0) 
 
   ! Mass of dry soil 
   call GetInput (MasSol,'MasSol','(g)',Valmin=0.d0) 
 
   ! Volume of liquid in moist soil and volume of liquid added 
   call GetInput (VolLiqSol,'VolLiqSol','(mL)',Valmin=0.d0) 
   call GetInput (VolLiqAdd,'VolLiqAdd','(mL)',Valmin=0.d0) 

   ! Calculate the volume of the suspension 
   VolLiqSus = VolLiqSol + VolLiqAdd 
 
   ! Organic matter content 
   call GetInput (CntOm,'CntOm','(kg.kg-1)',ValMin=0.d0) 
 
   ! End time 
   call GetInput (TimeEnd,'TimEnd','(d)',ValMin=0.d0) 
 
   ! Time step 
   call GetInput (DelTim,'DelTim','(d)',ValMin=0.d0) 
 
   StepsToPrint = max(1,int((DelTimPrint+1.d-10)/DelTim)) 
 
   ! Sorption parameters 
   !-------------------- 
 
   ! Reference concentration 
   call GetInput (ConLiqRef,'ConLiqRef','(mg.L-1)',ValMin=0.1d0) 
 
   ! Freundlich N 
   call GetInput (ExpFre,'ExpFre','(-)',ValMin=0.01d0,ValMax=1.3d0) 
 
   ! Equilibrium Kom 
   call GetInput (KomEql,'KomEql','(L.kg-1)',ValMin=0.d0) 
   CofFreEql = KomEql * CntOm 
 
   ! Ratio Kf,neq/Kf,eq 
   call GetInput (FacSorNeqEql,'FacSorNeqEql','(-)',Valmin=0.d0) 
   CofFreNeq = FacSorNeqEql * CofFreEql 
 
   ! Desorption rate coefficient 
   call GetInput (CofRatDes,'CofRatDes','(d-1)',ValMin=0.d0,ValMax=0.5d0) 
 
   ! Transformation parameters 
   !-------------------------- 
 
   ! Molar activation energy 
   call GetInput & 
   (MolEntTra,'MolEntTra','(kJ.mol-1)',ValMin=0.d0,ValMax=200.d0) 
   MolEntTra = 1.d3*MolEntTra 
 
   ! Pesticide half-life 
   call GetInput (DT50Ref,'DT50Ref','(d)',ValMin=1.d-1,ValMax=1.d6) 
 
   ! Reference temperature 
   call GetInput (TemRefTra,'TemRefTra','(C)') 
   TemRefTra = TemRefTra + 273.15d0 
 
   ! Experimental temperatures 
   !-------------------------- 
   call GetInput (TemTab,'Tem','(C)',Col=1) 
 
 
   ! Main part of programme 
   !----------------------- 
 
   write (FilOut,'(a)') ' ' 
   write (FilOut,'(a)') '--------------------------------------------------------------' 
   write (FilOut,'(a)') '* System properties' 
   write (FilOut,'(a,f10.4)') '* Mass of dry soil (g)                  :',MasSol 
   write (FilOut,'(a,f10.4)') '* Volume of water in moist soil (mL)    :',VolLiqSol 
   write (FilOut,'(a,f10.4)') '* Volume of water added (mL)            :',VolLiqAdd 
   write (FilOut,'(a,f10.4)') '* Initial mass of pesticide (ug)        :',MasIni 
   write (FilOut,'(a,f10.4)') '* Reference concentration (ug.mL-1)     :',ConLiqRef 
   write (FilOut,'(a,f10.4)') '* Equilibrium sorption coeff (mL.g-1)   :',CofFreEql 
   write (FilOut,'(a,f10.4)') '* Non-equili. sorption coeff (mL.g-1)   :',CofFreNeq 
   write (FilOut,'(a,f10.4)') '* Freundlich exponent (-)               :',ExpFre 
   write (FilOut,'(a,f10.4)') '* Desorption rate coefficient (d-1)     :',CofRatDes 
   write (FilOut,'(a,f10.4)') '* Half-life transformation (d)          :',DT50Ref 

   write (FilOut,'(a,f10.4)') '* Reference temperature (K)             :',TemRefTra 
   write (FilOut,'(a)') '--------------------------------------------------------------' 
   write (FilOut,'(a)') ' ' 
 
   write (FilOut,'("*",a6,1x,a6,3(1x,a20))') 'Temp','Time','Mas','ConLiq','XNeq' 
   write (FilOut,'("*",a6,1x,a6,3(1x,a20))') '(C)','(d)','(ug)','(ug.mL-1)','(ug.g-1)' 
 
   Temperatures: do T = 1,TemTab%NumRow 
    
      ! Calculate the coefficient at ambient temperature 
      Tem = TemTab%Y(1,T) + 273.15d0 
      FacTem = exp((-MolEntTra/RGas)*((1.d0/Tem)-(1.d0/TemRefTra))) 
 
      CofRatTra = FacTem*log(2.d0)/DT50Ref 
 
      ! Initialize the time loop 
      Mas = MasIni 
      XNeq = 0.d0 
 
      Tim = TimeStart 
      Steps = 0 
      TimeLoop: do 
 
         MasEql = Mas - MasSol*XNeq 
         ConPor = Freundlich (MasEql,MasSol,VolLiqSol,CofFreEql,ConLiqRef,ExpFre) 
         ConSus = Freundlich (MasEql,MasSol,VolLiqSus,CofFreEql,ConLiqRef,ExpFre) 
 
         if ((mod(Steps,StepsToPrint)) == 0) then 
            ! ConSus is always the only output concentration allowing direct fits of 
desorption 
   ! measurements 
   write (FilOut,'(2(1x,f6.1),3(1x,f20.8))') Tem-
273.15d0,Tim,Mas,ConSus,XNeq 
         end if 
 
         ! Integration of total mass 
         Mas = Mas + DelTim * (-1.d0*CofRatTra*(Mas-MasSol*XNeq)) 
 
         ! Integration of non-equilibrium domain 
         XNeq = XNeq + DelTim * & 
         (CofRatDes*(CofFreNeq*ConLiqRef*(ConPor/ConLiqRef)**ExpFre-XNeq)) 
 
         ! Increase time 
         Tim = Tim + DelTim 
         Steps = Steps + 1 
 
         if (Tim .dge. TimeEnd) exit 
 
      end do TimeLoop 
 
      write (FilOut,'(2(1x,f6.1),3(1x,f20.8))') Tem-273.15d0,Tim,Mas,ConSus,XNeq 
 
   end do Temperatures 
 
!========================================================================================= 
!========================================================================================= 
 
contains 
 
   double precision function Freundlich (Mas,MasSol,VolLiq,CofFreEql,ConLiqRef,ExpFre) 
   ! This function calculates the equilibrium concentration in a system 
   !-------------------------------------------------------------------------------------- 
 
      implicit none 
 
      double precision, parameter :: Err=1.d-4 
 
      double precision :: ConLiqOld,ConLiq,CofFre 
      double precision, intent(in) :: Mas,MasSol,VolLiq,CofFreEql,ConLiqRef,ExpFre 

      ConLiq=ConLiqRef 
      do 
         ConLiqOld = ConLiq 
         CofFre = & 
         CofFreEql * ConLiqRef**(1.d0-ExpFre) * (max(ConLiq,1.d-30) )**(ExpFre-1.d0) 
         ConLiq=Mas/(VolLiq+MasSol*CofFre) 
         if (abs(ConLiq-ConLiqOld) < Err*abs(ConLiq)) exit 
      end do 
 
      Freundlich = ConLiq 
 
   end function Freundlich 
 
!========================================================================================= 
!========================================================================================= 
 
   subroutine SetModelStamp () 
   ! Set the model stamp 
   !====================================================================================== 
 
      implicit none 
 
      Model%ExtInp = '.neq' 
      Model%ExtOut = '.out' 
      Model%ExtLog = '.log' 
      Model%ExtErr = '.err' 
      Call InitCh (Model%Date) 
      Call InitCh (Model%PearlVersion) 
      Call InitCh (Model%GeoVersion) 
      Call InitCh (Model%FocusVersion) 
      Model%PearlVersion = '3.3.3' 
      Model%Date = '10-May-2007' 
 
   end subroutine SetModelStamp 
 
!========================================================================================= 
!========================================================================================= 
 
   subroutine OpenPearlNeqFiles (ProgramPath) 
   ! Performs the following tasks: 
   ! (1) Opens the input and output files 
   ! (2) Prints the date-and-time and the Run Id to all opened output files. 
   ! (3) Reads the start-time and end-time, gets the print interval 
   ! (5) Sets the begin CPU time in seconds 
   ! 
   ! The following input and output files are used by the model: 
   ! Unit FilInp:   The input file (extension prl) 
   ! Unit FilOut:   The output file (extension out) 
   ! Unit FilLog:   The log file (extension log) 
   !====================================================================================== 
 
      implicit none 
 
      ! Declaration of local variables 
      !------------------------------- 
      character (len=LineLength) :: InFile,OutFile,LogFile,SumFile,ErrFile,RunName 
      integer :: IOS 
      character (len=WordLength) :: DateVal,TimeVal,ZoneVal 
      character (len=LineLength) :: ProgramName,ProgramPath 
      integer :: F 
      integer, dimension(8) :: TimArray 
 
      ! Main part of procedure 
      !----------------------- 
 
      ! Create Memory Space for the Words variable 
      Words%Allocated = .false. 
      Call Create (Words,NumWords) 
 
      ! Date and time 
      Call Date_And_Time (Date=DateVal,Time=TimeVal,Zone=ZoneVal,Values=TimArray) 

      ! Get the run ID 
      Call InitCh (RunName) 
      RunName = GetRun() 
 
      ! Get the path for the program 
      Call GetProgramName (ProgramName) 
      Call GetPath (ProgramName,ProgramPath) 
 
      ! Construct the file names (add the extensions) 
      call InitCh (InFile) 
      call InitCh (OutFile) 
      call InitCh (LogFile) 
      call InitCh (SumFile) 
      call InitCh (ErrFile) 
      InFile  = trim(RunName)//Model%ExtInp 
      OutFile = trim(RunName)//Model%ExtOut 
      LogFile = trim(RunName)//Model%ExtLog 
      ErrFile = trim(RunName)//Model%ExtErr 
 
      ! Open the input file 
      Open (FilInp,file=trim(InFile),status='old',IOStat=IOS) 
      if (IOS /= 0) then 
         ! Error condition - abort program execution 
         Error%Code = -1 
         write ( Error%m1,'("Cannot find file ",a," with status old")') trim(InFile) 
         stop 'Illegal run id - no error file generated' 
      end if 
      rewind (FilInp) 
 
      ! Open the error file 
      Call OpenAfterDelete (FilErr,trim(ErrFile)) 
 
      ! Open the output file 
      Call OpenAfterDelete (FilOut,trim(OutFile)) 
 
      ! Open the log file 
      Call OpenAfterDelete (FilLog,trim(LogFile)) 
 
      write (*,'("* ")') 
      write (*,'("* ------------------------------------------------------------")') 
      write (*,'("* PEARLNEQ (c) MNP/RIVM/Alterra")') 
      write (*,'("* ------------------------------------------------------------")') 
      write (*,'("*")') 
      write (*,'("* PEARLNEQ version 4")') 
      write (*,'("* PEARLNEQ created on ",a)') trim (Model%Date) 
      write (*,'("* ")') 
      write (*,'("* ------------------------------------------------------------")') 
      write (*,'("* ")') 
 
      ! Write the Run ID, file-names and date-and-time to the output file 
      do F = 21,22 
         write (F,'("* ------------------------------------------------------------& 
         &------------------")') 
         write (F,'("* Results from PEARLNEQ (c) MNP/RIVM/Alterra")') 
         write (F,'("* PEARLNEQ version 4")') 
         write (F,'("* PEARLNEQ created on ",a)') trim (Model%Date) 
         write (F,'("* ")') 
         write (F,'("* Run ID                                       : ",a)') trim 
(GetRun()) 
         write (F,'("* Input file generated on                      : ",a2,"-",a2,"-",a4)') 
& 
         DateVal(7:8),DateVal(5:6),DateVal(1:4) 
         write (F,'("* ------------------------------------------------------------& 
         &------------------")') 
         write (F,'("* ")')  
 
      end do 
   end subroutine OpenPearlNEQFiles 
end program PearlNeq 

